ECONOMICS 531
PROJECT 1

Profit Maximization with a Unilateral Externality

This project has three parts. In Part 1 we will solve a profit-maximization problem for a price-taking firm. We will do this using three different approaches. The first approach breaks the problem into two stages: a cost-minimization stage in which we derive the cost function (a constrained optimization problem); and a profit-maximization stage using the cost function from stage one (an unconstrained optimization problem).

The second approach solves the profit-maximization problem directly without first deriving the cost function.

The third approach solves the profit-maximization problem in a way that allows the derivation of marginal private benefit and marginal private cost functions for the firm; these will be used in Part 2 of the project.

In Part 2 we will introduce a second firm, whose output is degraded by a production input used by the firm from Part 1. That is, we introduce a unilateral externality. We will solve the profit-maximization problem for the damaged firm, and use the solution to derive a marginal external cost function. We will then combine this with the marginal private cost function and marginal private benefit function for the firm from Part 1 to find the social optimum in this setting.

In Part 3 we will solve a joint-profit-maximization problem (in which the sum of profits for the two firms are maximized). We will then relate this solution to the social optimum derived in Part 2.

The entire project should be done in **Maple**. An appendix to this document provides the code for the first part of Part 1 (up to equation (6)). Start with this code and build on it for the rest of the project.

PART 1

Plotting the Production Function

Consider a price-taking firm that produces output y using two inputs x_1 and x_2 in a production process described by the following production function:

(1) $$f(x_1, x_2) = \alpha_1 x_1^{1/2} + \alpha_2 x_2^{1/2}$$

Plot this production function in three dimensions using $\alpha_1 = \frac{1}{2}$ and $\alpha_2 = \frac{1}{2}$ over the range $x_1 \in [0,3]$ and $x_2 \in [0,3]$. Label this **Figure 1**.

Now create a plot of isoquants using the *implicitplot* command over the range $x_1 \in [0,3]$ and $x_2 \in [0,3]$ for four different levels of output: $y = 3/4$, $y = 1$, $y = 5/4$ and $y = 2$. Label this **Figure 2**. Explain why the isoquant for $y = 2$ does <u>not</u> show up in your plot even though you asked *Maple* to display it. (The reason relates to the range specified for x_1 and x_2).

We now want to solve the profit-maximization problem for this firm. We will use three different approaches to the problem.

First Approach: The Two-Stage Problem

Here we solve the profit-maximization problem in two stages. In the first stage we derive the cost function as the solution to a cost-minimization problem. In the second stage we use the cost function to express profit as a function of output, and then choose output to maximize profit.

<u>Stage 1: The Cost-Minimization Problem</u>

The problem is to minimize the cost of producing a given level of output, y. The firm faces input prices w_1 and w_2 for inputs x_1 and x_2 respectively. Its cost-minimization problem is therefore given by

(2) $$\min_{x_1 x_2} \; w_1 x_1 + w_2 x_2 \quad \text{subject to} \quad y = \alpha_1 x_1^{1/2} + \alpha_2 x_2^{1/2}$$

Create a graph depicting this problem in (x_1, x_2) space over the range $x_1 \in [0,3]$ and $x_2 \in [0,3]$ by plotting an isoquant drawn for $y = 1$, and three iso-cost contours at cost values \$3, \$4 and \$5. Use $\alpha_1 = \frac{1}{2}$, $\alpha_2 = \frac{1}{2}$, $w_1 = 2$ and $w_2 = 2$ for your plot. Label this **Figure 3**. Based on your graph, approximate the optimal values of x_1 and x_2.

Write down the Lagrangean (using λ to denote the multiplier) and derive the three first-order conditions.

Solve the three order-order conditions simultaneously, and confirm that you obtain the following solutions for the conditional input demands:

(3) $$x_1(y,w) = \left(\frac{\alpha_1 w_2 y}{\alpha_1^2 w_2 + \alpha_2^2 w_1} \right)^2$$

(4) $$x_2(y,w) = \left(\frac{\alpha_2 w_1 y}{\alpha_1^2 w_2 + \alpha_2^2 w_1} \right)^2$$

(5) $$\lambda(y,w) = \frac{2 w_1 w_2 y}{\alpha_1^2 w_2 + \alpha_2^2 w_1}$$

Evaluate (3) and (4) at $\alpha_1 = \frac{1}{2}$, $\alpha_2 = \frac{1}{2}$, $w_1 = 2$, $w_2 = 2$ and $y = 1$. These solutions should accord with your approximation of the optimal values from your **Figure 3**.

Construct the cost function and confirm that it is

(6) $$c(y,w) = \frac{w_1 w_2 y^2}{\alpha_1^2 w_2 + \alpha_2^2 w_1}$$

Derive the marginal cost function and relate your answer to the value of the Lagrange multiplier from (5) above.

<u>Stage 2: The Profit-Maximization Problem</u>

The firm faces a fixed price so its profit-maximization problem is

(7) $$\max_{y} \quad py - c(y, w)$$

Solve the first order-order condition and confirm that you obtain the following solution for the supply function:

(8) $$y(p,w) = \frac{p(\alpha_1^2 w_2 + \alpha_2^2 w_1)}{2w_1 w_2}$$

Substitute the supply function for y in the conditional input demands from (3) and (4), and confirm that you obtain the following solutions for the input demands:

(9) $$x_1(p,w) = \frac{p^2 \alpha_1^2}{4w_1^2}$$

(10) $$x_2(p,w) = \frac{p^2 \alpha_2^2}{4w_2^2}$$

Construct the profit function and confirm that is

(11) $$\pi(p,w) = \frac{p^2(\alpha_1^2 w_2 + \alpha_2^2 w_1)}{4w_1 w_2}$$

Second Approach: Direct Profit-Maximization

The direct profit-maximization problem is expressed directly in terms of input choices:

(12) $$\max_{x_1, x_2} \quad pf(x_1, x_2) - w_1 x_1 - w_2 x_2$$

Plot this objective function in three dimensions using $\alpha_1 = \frac{1}{2}$, $\alpha_2 = \frac{1}{2}$, $w_1 = 2$, $w_2 = 2$ and $p = 8$ over the range $x_1 \in [0,3]$ and $x_2 \in [0,3]$. Label this **Figure 4**. Note that it has a free turning point (or unconstrained optimum). This reflects the fact that the production function is strictly concave.

Solve the first order-order conditions and confirm that you obtain the following solutions for the input demands:

(13) $$x_1(p,w) = \frac{p^2\alpha_1^2}{4w_1^2}$$

(14) $$x_2(p,w) = \frac{p^2\alpha_2^2}{4w_2^2}$$

Note that these are the same factor demands that we derived in (9) and (10) from the two-stage problem. There we derived them in the second stage of a two-stage problem; here we derived them directly.

Substitute these factor demands for x_1 and x_2 in the production function from (1), and confirm that you obtain

(15) $$y(p,w) = \frac{p(\alpha_1^2 w_2 + \alpha_2^2 w_1)}{2w_1 w_2}$$

This is the same supply function that we derived in (8) from the two-stage problem.

Now substitute the factor demands back into the objective function from (12) to derive the value function for this problem, and confirm that it is given by

(16) $$\pi(p,w) = \frac{p^2(\alpha_1^2 w_2 + \alpha_2^2 w_1)}{4w_1 w_2}$$

This is of course the same profit function that we derived in (11) from the two-stage problem.

Third Approach: Marginal Cost and Marginal Benefit of Using an Input

We now want to think again of the profit-maximization problem as a two-stage problem, but in a way different from our first approach to the problem. In particular, we want to conduct the following thought experiment. First imagine that the firm cannot choose the amount of x_2 it uses; we can think of it for the moment as a fixed input. It chooses only the amount of x_1, taking as given the fixed amount of x_2. We can solve the profit-maximizing problem for x_1, and derive the associated revenue and cost as functions of x_2. We can then allow x_2 to be chosen in the second stage of the overall problem. This will allow us to cast the profit-maximization problem in terms of the marginal private benefit (revenue) and marginal private cost of using x_2. This in turn will allow us to frame the problem in the familiar language of unilateral externalities.

<u>Stage 1: The Choice of Input 1</u>

If x_2 is fixed at some given level then the choice problem for the firm is simply a choice over x_1:

(17) $$\max_{x_1} \; pf(x_1, x_2) - w_1 x_1 - w_2 x_2$$

Solve the first order-order condition and confirm that you obtain the following solution:

(18) $$x_1(p, w) = \frac{p^2 \alpha_1^2}{4 w_1^2}$$

Note that this input demand is <u>not</u> a function of x_2 in this case. This reflects the fact that for the simple production function we have used for this exercise, the marginal productivity of x_1 is independent of x_2. This would not be true of a more general production function. However, the simple one we have used here suits our purposes by keeping the mathematics relatively simple.

We can now use (18) to derive expressions for revenue and cost as functions of x_2 when x_1 is chosen optimally. In particular, confirm that revenue is given by

(19) $$r(x_2) = \frac{p^2 \alpha_1^2}{2 w_1} + p \alpha_2 x_2^{1/2}$$

and that cost is given by

(20) $$c(x_2) = \frac{p^2 \alpha_1^2}{4 w_1} + w_2 x_2$$

Stage 2: The Choice of Input 2

We can think of revenue, as described by (19), as the private benefit the firm derives from using x_2. Similarly, we can think of cost, as described by (20), as the private cost to the firm of using x_2.

Accordingly, confirm that marginal private benefit (*MPB*) for the firm is

(21) $$MPB(x_2) \equiv \frac{\partial r(x_2)}{\partial x_2} = \frac{p\alpha_2}{2x_2^{1/2}}$$

and that marginal private cost (*MPC*) for the firm is

(22) $$MPC(x_2) \equiv \frac{\partial c(x_2)}{\partial x_2} = w_2$$

Plot these two functions using $\alpha_2 = \frac{1}{2}$, $w_2 = 2$ and $p = 8$ over the range $x_2 \in [\frac{1}{4}, 2]$. Label this **Figure 5**.

From the plot we can see that *MPB* crosses *MPC* from above, so we know that the privately-optimal choice for the firm is where $MPB(x_2) = MPC(x_2)$. Confirm that the solution to this condition is

(23) $$x_2(p, w) = \frac{p^2\alpha_2^2}{4w_2^2}$$

Note that this is the same factor demand that we derived in (10) and (14) from our earlier approaches to the profit-maximization problem. Expressing the problem in terms *MPB* and *MPC* does not change the fundamental nature of the decision problem; it just provides a different way of looking at it.

We can also usefully express the problem in terms of marginal net private benefit. In particular, we can define marginal net private benefit as

(24) $$MNPB(x_2) = MPB(x_2) - MPC(x_2)$$

Then the privately optimal choice for the firm is where $MNPB(x_2) = 0$.

Plot $MNPB(x_2)$ using $\alpha_2 = \frac{1}{2}$, $w_2 = 2$ and $p = 8$ over the range $x_2 \in [\frac{1}{4}, 2]$. Label this **Figure 6**.

In the next part of the project we will introduce an externality associated with the use of x_2, and show that the privately-optimal choice for the firm is not socially optimal.

PART 2

In this part we introduce an externality, using the terminology from Chapter 2.

In this part we will refer to the firm from Part 1 as “firm A”, and introduce a second firm, “firm B”, that produces output z using inputs x_3 and x_4 according to the following production function:

(25) $$g(x_3, x_4; x_2) = \beta_1 x_3^{1/2} + \beta_2 x_4^{1/2} - \delta x_2^{3/2}$$

where x_2 is the amount of input 2 used by firm A, and δ is the “damage parameter”. Thus, the use of input 2 by firm A imposes a negative externality on firm B via its deleterious impact on output. (Think of firm A as a polluting factory, and firm B as a nearby farm).

Suppose firm B faces price q for its output, and pays prices w_3 and w_4 for inputs x_3 and x_4 respectively. It can freely choose its use of x_3 and x_4 but it has no control over the use of x_2 by firm A.

Confirm that the profit function for firm B is

(26) $$\pi_B(q, w; x_2) = \frac{q^2(\beta_1^2 w_4 + \beta_2^2 w_3)}{4 w_3 w_4} - \delta q x_2^{3/2}$$

Evaluate this profit for firm B at the privately-optimal choice of x_2. Let $\hat{\pi}_B$ denote this profit.

The external cost imposed on firm B by firm A is loss of profit for firm B caused by the use of x_2 by firm A. Confirm that the marginal external cost is

(27) $$MEC(x_2) \equiv -\frac{\partial \pi_B(q,w)}{\partial x_2} = \frac{3\delta q x_2^{1/2}}{2}$$

Given this externality, what is the socially-optimal use of x_2 by firm A?

Approach this question in two equivalent ways. First, construct the marginal social cost (*MSC*) function:

(28) $$MSC(x_2) = MPC(x_2) + MEC(x_2)$$

Plot $MSC(x_2)$ using $\delta = \frac{1}{10}$ and $q = 8$ over the range $x_2 \in [\frac{1}{4}, 2]$, overlaid on **Figure 5**. Label this combined plot **Figure 7**.

Assuming that there is no positive externality associated with the use of x_2 (which would otherwise cause marginal private benefit and marginal social benefit to diverge), the socially-optimal level of use is x_2^*, characterized by

(29) $$MSC(x_2^*) = MPB(x_2^*)$$

Confirm that

(30) $$x_2^* = \frac{w_2(2w_2 - 2(w_2^2 + 3\delta\alpha_2 pq)^{\frac{1}{2}}) + 3\delta\alpha_2 pq}{(3\delta q)^2}$$

Warning. *Maple* will generate two solutions to (29). Make sure you pick the correct one. (Try assigning some parameter values and comparing the solutions with your **Figure 7**).

Evaluate x_2^* at $p = 0$, and explain the result.

Now derive x_2^* in a different way. Plot $MEC(x_2)$ using $\delta = \frac{1}{10}$ and $q = 8$ over the range $x_2 \in [\frac{1}{4}, 2]$, overlaid on **Figure 6**. Label this combined plot **Figure 8**. Confirm that the solution to $MEC(x_2) = MNPB(x_2)$ is x_2^*, as described in (30) above, and explain why we obtain the same result either way.

Let $\hat{x}_2$ denote the privately-optimal choice of x_2, as given by (23). The social surplus foregone at this level of input, relative to the social optimum, is

(31) $$L = \int_{x_2^*}^{\hat{x}_2} (MSC(x_2) - MPB(x_2))dx_2$$

This is the area in **Figure 7** bounded by the *MSC* schedule and the *MPB* between x_2^* and $\hat{x}_2$. (See **Figure 2-34** from the lecture notes for a comparable figure).

Equivalently, this foregone surplus can be calculated as

(32) $$L = \int_{x_2^*}^{\hat{x}_2} (MEC(x_2) - MNPB(x_2))dx_2$$

This is the area in **Figure 8** bounded by the *MEC* schedule and the *MNPB* between x_2^* and $\hat{x}_2$. (See **Figure 2-38** from the lecture notes for a comparable figure).

Confirm that (31) and (32) do indeed yield the same answer using the following parameter values: $\alpha_1 = \frac{1}{2}$, $\alpha_2 = \frac{1}{2}$, $w_1 = 2$, $w_2 = 2$, $p = 8$, $\beta_1 = \frac{1}{2}$, $\beta_2 = \frac{1}{2}$, $w_3 = 2$, $w_4 = 2$, $q = 8$ and $\delta = \frac{1}{10}$.

PART 3

In principle, the two firms should be able to internalize the externality by acting cooperatively to maximize their joint profit. Here we will examine that problem.

The joint-profit maximization problem is

(33) $$\max_{x_1,x_2,x_3,x_4} \; pf(x_1, x_2) + qg(x_3, x_4) - w_1x_1 - w_2x_2 - w_3x_3 - w_4x_4$$

Derive the four first-order conditions for this problem.

Solve the first-order conditions simultaneously, and confirm that your solution for x_2 is x_2^* from (30) above. Is this a coincidence? Warning. Maple will return a solution of the form "root of" here. You will probably need help on how to proceed.

Calculate the maximized joint profit using the parameter values from the last question in Part B. Let π_J^* denote the result. Using those same parameter values, calculate profit for each firm when the firms act independently and firm A uses its privately optimal level of x_2. Let $\hat{\pi}_A$ and $\hat{\pi}_B$ denote these profits for firms A and B respectively.

Calculate $\pi_J^* - (\hat{\pi}_A + \hat{\pi}_B)$, and explain the relationship between this value and the foregone-surplus value we calculated in the last question from Part 2.

531 PROJECT 1

```
> restart:
> with(plots):
```

Warning, the name changecoords has been redefined

PART 1

Plotting the Production Function

Specifiy the function:

```
> f:=alpha1*(x1)^(1/2)+alpha2*(x2)^(1/2);
```

$$f := \alpha 1 \sqrt{x1} + \alpha 2 \sqrt{x2}$$

Choose parameter values:

```
> alpha1:=1/2:
> alpha2:=1/2:
```

and plot in 3D:

```
> plot3d(f,x1=0..3,x2=0..3,axes=boxed);
```



FIGURE 1

Plot in 2D:

```
> g1:=implicitplot(f=3/4,x1=0..3,x2=0..3,color=blue):
> g2:=implicitplot(f=1,x1=0..3,x2=0..3,color=blue):
```

```
> g3:=implicitplot(f=5/4,x1=0..3,x2=0..3,color=blue):
> g4:=implicitplot(f=2,x1=0..3,x2=0..3,color=blue):
> display(g1,g2,g3,g4);
```



FIGURE 2

Reset parameters:

```
> alpha1:='alpha1':
> alpha2:='alpha2':
```

First Approach: The Two-Stage Problem

Stage 1: The Cost-Minimization Problem

Specify cost:

```
> c:=w1*x1+w2*x2;
```

$$c := w1\ x1 + w2\ x2$$

Choose parameter values for plotting:

```
> alpha1:=1/2:
> alpha2:=1/2:
> w1:=2:
> w2:=2:
```

Plot an isoqant and three isocost contours:

```
> g1:=implicitplot(c=3,x1=0..3,x2=0..3):
```

```
> g2:=implicitplot(c=4,x1=0..3,x2=0..3):
> g3:=implicitplot(c=5,x1=0..3,x2=0..3):
> g4:=implicitplot(f=1,x1=0..3,x2=0..3,colour=blue):
> display(g1,g2,g3,g4);
```



FIGURE 3

Based on Figure 3, the solution at these parameter values appears to be x1=1 and x2=1.

Reset parameters:

```
> alpha1:='alpha1':
> alpha2:='alpha2':
> w1:='w1':
> w2:='w2':
```

Construct the Lagrangean:

```
> L:=c+lambda*(y-f);
```

$$L := w1\ x1 + w2\ x2 + \lambda\,(y - \alpha 1\,\sqrt{x1} - \alpha 2\,\sqrt{x2}\,)$$

Construct the FOCs:

```
> foc1:=diff(L,x1);
```

$$foc1 := w1 - \frac{\lambda\,\alpha 1}{2\sqrt{x1}}$$

```
> foc2:=diff(L,x2);
```

$$foc2 := w2 - \frac{\lambda\,\alpha 2}{2\sqrt{x2}}$$

```
> foc3:=diff(L,lambda);
```

$$foc3 := y - \alpha 1\sqrt{x1} - \alpha 2\sqrt{x2}$$

Solve these three equations:

```
> soln:=solve({foc1,foc2,foc3},{x1,x2,lambda});
```

$$soln := \left\{x2 = \frac{w1^2\,y^2\,\alpha 2^2}{(w1\,\alpha 2^2 + w2\,\alpha 1^2)^2},\ x1 = \frac{w2^2\,y^2\,\alpha 1^2}{(w1\,\alpha 2^2 + w2\,\alpha 1^2)^2},\ \lambda = \frac{2\,w1\,w2\,y}{w1\,\alpha 2^2 + w2\,\alpha 1^2}\right\}$$

Label the solutions for future reference:

```
> x1tilde:=subs(soln,x1);
```

$$x1tilde := \frac{w2^2\,y^2\,\alpha 1^2}{(w1\,\alpha 2^2 + w2\,\alpha 1^2)^2}$$

```
> x2tilde:=subs(soln,x2);
```

$$x2tilde := \frac{w1^2\,y^2\,\alpha 2^2}{(w1\,\alpha 2^2 + w2\,\alpha 1^2)^2}$$

```
> lambdatilde:=subs(soln,lambda);
```

$$lambdatilde := \frac{2\,w1\,w2\,y}{w1\,\alpha 2^2 + w2\,\alpha 1^2}$$

The relationship between x1 and w2:

```
> simplify(diff(x1tilde,w2));
```

$$\frac{2\,w2\,y^2\,\alpha 1^2\,w1\,\alpha 2^2}{(w1\,\alpha 2^2 + w2\,\alpha 1^2)^3}$$

Positive. An increase in w2 induces a <u>substitution</u> out of x2 and into x1.

Evalaute the factor demands at the plotted parameters values:

```
> evalf(subs(alpha1=1/2,alpha2=1/2,w1=2,w2=2,y=1,x1tilde));
```

$$1.$$

```
> evalf(subs(alpha1=1/2,alpha2=1/2,w1=2,w2=2,y=1,x2tilde));
```

$$1.$$

Construct the value function (the cost function):

```
> C:=simplify(subs(x1=x1tilde,x2=x2tilde,c));
```

$$C := \frac{w1\,w2\,y^2}{w1\,\alpha 2^2 + w2\,\alpha 1^2}$$

The marginal cost function:

```
> MC:=diff(C,y);
```

$$MC := \frac{2\, w1\, w2\, y}{w1\, \alpha 2^2 + w2\, \alpha 1^2}$$

In comparison, the Lagrange multiplier (LM):

```
> lambdatilde;
```

$$\frac{2\, w1\, w2\, y}{w1\, \alpha 2^2 + w2\, \alpha 1^2}$$

They are equal, as a consequence of the envelope thereom. The LM in this problem measures the increase in cost when the constraint is tightened (output is increased) by a marginal amount.